
Abstract. We use the graphics processing unit (GPU) for fast calculations of helicity amplitudes of physics
processes. As our first attempt, we compute uu → nγ (n=2 to 8) processes in pp collisions at

√
s = 14TeV

by transferring the MadGraph generated HELAS amplitudes (FORTRAN) into newly developed HEGET
(HELAS Evaluation with GPU Enhanced Technology) codes written in CUDA, a C-platform developed
by NVIDIA for general purpose computing on the GPU. Compared with the usual CPU programs, we
obtain 40-150 times better performance on the GPU.

1 Introduction

1.1 Physics motivations

The field of particle physics is about to enter a new era
of experimental discovery at the Large Hadron Collider
(LHC) which will start collecting data this year. Typical
new physics signals at the LHC are expected to have many
high pT jets, γ’s, W ’s and Z’s, and it is very important to
estimate the Standard Model (SM) background to all the
signals reliably. Computation of multiple particle produc-
tion amplitudes is, however, often time consuming, and it
is desirable to have tools that allow us to evaluate cross
sections efficiently. We examine here the possibility of us-
ing a GPU (Graphic Processing Unit) to compute helicity
amplitudes at many phase space points simultaneously, by
taking advantage of its parallel processing capability.

1.2 GPU hardware and software

The GPU is a specialized integrated circuit (IC) used in
the computer system which outputs complex images onto
displays very fast. This IC is composed of many multi-
processors and can process large amounts of image data
in parallel with high efficiency. Because of this high perfor-
mance GPUs are widely used in the scientific applications
where a large number of calculations must be performed
to process the data.

a e-mail: junichi.kanzaki@kek.jp
b e-mail: naotoshi@post.kek.jp
c e-mail: tstelzer@uiuc.edu

Recently NVIDIA [1] introduced the software devel-
opment system, CUDA[2]1, which enables one to develop
programs which can be executed on the GPU using C/C++.
We use this system to develop a new helicity amplitude
calculation package which can be used on a GPU.

2 Physics process

As our first attempt we compute uu → nγ processes in
pp-collisions, because the amplitudes are particularly sim-
ple for numerical codes like HELAS [3] such that large
numbers of Feynman diagrams, beyond the capability of
automatic diagram generators like MadGraph [4], can be
computed without difficulty [5].

2.1 nγ production in pp collisions

The cross section for producing n high pT photons in pp
collisions can be expressed in the leading order of pertur-
bative QCD as

dσnγ =
∑

q

∫∫

dx1dx2
[

Dq/p (x1, Q)Dq̄/p (x2, Q)

+ Dq̄/p (x1, Q)Dq/p (x2, Q)
]

dσ̂ (ŝ; y) ,(1)

where the factorization scale Q for the q and q distribution
functions is chosen as the pT cut-off, and the qq → nγ’s
process cross section is

dσ̂ (ŝ) =
1

2ŝ

1

∑

∣Mλ1,··· ,λn

σ1,σ2

∣

∣

2 1

n!
dΦn . (2)

1 We use CUDA version 2.1 in this paper.

Here

ŝ = s x1x2 , (3)

and σi(i = 1, 2) denotes the helicity of the initial u and
u, λi(i=1 to n) is that of the final photons. The n-body
phase space is

dΦn = (2π)
4
δ4

(

p1 + p2 −
n
∑

i=1

ki

)

n
∏

i=1

d3ki
(2π)3 2ωi

, (4)

where ωi = |ki| for massless photons and 1/n! is the statis-
tical factor for n identical photons. The helicity amplitude
for the process

q(p1, σ1)+q(p2, σ2) → γ(k1, λ1)+γ(k2, λ2)+· · ·+γ(kn, λn)
(5)

is particularly simple:

Mλ1,··· ,λn

σ1,σ2
=Mλ1,··· ,λn

σ1,σ2
+ (n!− 1) permutations , (6)

where the full amplitude is obtained by (n!− 1) permuta-
tion of one amplitude, denoted e.g. by the Feynman dia-
gram at Fig. 1:

Mλ1,··· ,λn

σ1,σ2
= v̄(p2, σ2)

[

γµn

/p1 − /Kn−1

tn−1

γµn−1 · · ·

· · · γµ3
/p1 − /K2

t2
γµ2

/p1 − /K1

t1
γµ1

]

u(p1, σ1)

ǫ∗µ1
(k1, λ1) · · · ǫ∗µn

(kn, λn) (eQq)
n , (7)

with

Kl =

l
∑

i=1

ki, tl = (p1 −Kl)
2
. (8)

Here Qq is the electric charge of the quark in units of e,
the proton charge.

The standard version of MadGraph/MadEvent [4,6,
7] generates HELAS amplitudes and computes the cross
section for the nγ production process up to n = 6 with-
out difficulty. The HELAS amplitude for n = 7 with 7 !
diagrams can be generated by MadGraph after enlarging
the maximum size of several variables. We are unable to
make MadGraph generate 8 ! ≈ 4× 104 diagrams for n=8
photons, so we evaluate them by coding the amplitude of
eq. (6) and by summing over (8!−1) permutations of the
8 photon momenta in a numerical program.

q q̄

p1 p2

k1 k2 k3 kn

γ γ γ γ

Fig. 1. One of n! diagrams for qq → nγ.

2.2 Final state cuts

In order to simulate realistic LHC experiments, we intro-
duce final state cuts for all the observed photons as follows:

pTi > pcut
T

= 20GeV,

|ηi| < ηcut = 2.5,

∆Rij > ∆Rcut = 0.4,

(9)

where ηi is the rapidity of the i-th photon and

∆Rij ≡
√

(∆ηij)2 + (∆φij)2 (10)

measures rapidity and azimuthal angle separation between
two photons.

For the sake of simplicity and for the purpose of the
present paper, we consider only one subprocess, uu →
nγ′s, for the n-photon production processes at the LHC,
by neglecting contributions from dd, ss, cc and bb colli-
sions.

3 Computation on the GPU

On the GPU, a number of programs can be executed with
its multi-processors in parallel. These programs which run
concurrently must be the same for all multi-processors
with different input data for each program. In order to
utilize the high performance of the GPU for computing
the n-photon production processes, we assign a program
which processes one event on each processor.

If the program is given a set of random numbers, it
generates a phase space point and computes a Jacobian
and a squared amplitude for the phase space point inde-
pendently from the other programs. By keeping the in-
dependence of programs among processors their structure
can become simple, and we can take full advantage of the
highly parallel computation on the GPU.

In this section we describe our software developed for
the computation of the n-photon productions in pp colli-
sions on the GPU and its computation environment.

3.1 Structure of the program

For n-photon production processes our program generates
momenta and helicities of photons in the final state, those
of u and u in the initial state, and computes the total cross
section of the physics process in the following order:

1. initialization of the program,
2. random number generation on the CPU,
3. transfer random numbers to the GPU,
4. generate helicities and momenta of u, u and nγ’s using

random numbers, and compute squared amplitudes on
the GPU,

5. transfer the momenta and helicities of external parti-
cles, as well as computed squared amplitudes to the
CPU, and

Table 1. Parameters of GPUs

GeForce GeForce GeForce
GTX280 9800GTX 8800M GTS

Number of 30 16 8
multiprocessor
Number of core 240 128 64
Total amount of 1000 500 500

global memory [MB]
Total amount of 64 64 64

constant memory [kB]
Total amount of shared 16 16 16
memory per block [kB]

Total number of registers 16 8 8
available per block [kB]

Clock rate [GHz] 1.30 1.67 0.40

6. sum up all values to obtain the total cross section and
distributions on the CPU.

Program steps between the generation of random num-
bers (2) and the summation of computed cross sections (6)
are repeated until we obtain enough statistics for the cross
section and distributions.

For the computation of physics quantities for the n-
photon production processes, we prepare two types of am-
plitude programs. They are:

– a program converted from the FORTRAN program ob-
tained with the MadGraph [4], and

– a handwritten CUDA program using permutations of
final state photons, via eq. (6).

Both programs are written using the HEGET (HELAS
Evaluation with GPU Enhanced Technology) functions,
newly developed codes, written in CUDA [2], to compute
helicity amplitudes a la HELAS [3].

Sample codes for the case of the 3-photon production
process are shown in Appendix A as List 1 and List 2 for
the converted MadGraph amplitude and the permutation
amplitude, respectively. They compute a sum of ampli-
tudes for all diagrams, ampsum, a complex number, from
the given momenta, {p1, . . ., p5} and helicities, {nh1,
. . ., nh5} of the external particles. In the code for the
permutation amplitude, List 2, a function, iPNext, gen-
erates a set of integers for the next permutation in the
sequence. For all computation of amplitudes, new data
type “cmplx”, defined in List 16, for complex numbers on
GPU is used.

3.2 Execution of a CUDA function on GPU

Once a set of random numbers is generated on the CPU,
computations from the generation of phase space points to
the calculation of squared amplitudes are done by calling
a CUDA function which is executed on the GPU. A func-
tion, which is called from a CPU program and is executed
on a GPU, is called a kernel. In our program, a single call
of a kernel function computes the scattering amplitudes
at multiple phase-space points in parallel on a GPU.

A unit program, called a thread, which is executed on
a single processor, calculates the amplitudes for one event.
In the CUDA programming model a set of threads forms
a thread block. Threads within one thread block can share
data through the shared memory and cooperate among
themselves. In the current architecture of NVIDIA’s GPUs
a thread block can contain up to 512 threads. The size of
a thread block can be changed within this limit when the
program is executed on a GPU, and we optimize it to
obtain the best performance of our program.

With a single call of a kernel, multiple thread blocks
are executed in parallel on a GPU. A set of thread blocks,
which is executed with a single kernel call, is called a grid.
Even within a grid, threads in different thread blocks can-
not share data through the shared memory.

Table 2. Configuration of the host PC

Linux PC iMac

CPU Core2Duo 3GHz Core2Duo 3.06GHz
L2 Cache 6MB 6MB
Memory 4GB 2GB
Bus Speed 1.333GHz 1.07GHz

OS Fedora 8 (64 bit) MacOS X 10.5.5

GPU GTX280 & 8800M GTS
9800GTX

3.3 Host PC environment

We tested our program on three different GPUs by NVIDIA:
the GeForce GTX280, 9800GTX and 8800MGTS. The pa-
rameters for these three GPUs are given in Table 1. The
GeForce 9800GTX with 128 processors was introduced in
April/2008 by NVIDIA as a high-end single GPU graphic
card for the high performance output of complex images
like 3D to the PC display. The GeForce GTX280 with a
new processor architecture, introduced in June/2008, has
240 processors, 1GByte global memory and 16k registers,

Table 3. HEGET functions for external lines

External Line HEGET Function HELAS Subroutine

Flowing-In Fermion ixxxx0, ixxxx1, ixxxx2 IXXXXX

Flowing-Out Fermion oxxxx0, oxxxx1, oxxxx2 OXXXXX

Vector Boson vxxxx0, vxxxx1, vxxxx2 VXXXXX

Table 4. List of the HEGET vertex functions

Vertex Inputs Output HEGET Function HELAS Subroutine

FFV FFV Amplitude iovxxx IOVXXX

FF V jioxx0 JIOXXX

FV F fvixx0, fvoxx0 FVIXXX, FVOXXX

which are about twice those of 9800GTX, with a 20%
slower clock rate. A GeForce 8800M GTS is installed on
an Apple iMac, and has 64 processors and the clock rate
of 0.4GHz, but still has the same memory and register size
as the 9800GTX.

Parameters for the two host PCs are summarized in
Table 2. They have comparable capability, although the
Linux PC has twice larger memory size and 25% faster bus
speed. It should be noted here that although the host PC’s
both have two CPU’s our programs do not run parallel on
them.

4 HEGET functions

Based on the FORTRAN version of the HELAS library[3]
we developed a set of CUDA functions which can be used
on a GPU for performing helicity amplitude calculations.
These functions are directly converted from the HELAS
subroutines into C code. We kept the order of the ar-
guments of the HELAS subroutines but parameters for
masses and widths were removed from the argument list.

The HEGET functions are organized to maximize the
performance on the GPU. Since conditional branches with-
in programs using “if” statements reduces the total effi-
ciency of parallel computing, we eliminated them as much
as possible in the HEGET functions. Accordingly, we pre-
pared separate HEGET codes for the computation of mas-
sive and massless wave functions. In this paper we only use
the HEGET functions for massless particles.

By the same token, we introduce three types of func-
tions for external wave functions; “1” for particles moving
in the +z direction, “2” for particles moving in the −z
direction and “0” for particles with momentum, p, along
a generic direction. These numbers are appended to the
name of the corresponding functions.

All functions relevant to this paper are listed in Ta-
ble 3 and Table 4, which include all functions for massless
fermions and massless vector bosons (photons), respec-
tively. The naming scheme for HEGET functions follow
that of HELAS subroutines: the HEGET (HELAS) func-
tion names start with i(I) and o(O) for flow-in and flow-
out fermion wave functions, respectively, v(V) for vector
boson wave functions, f(F) for off-shell fermions and j(J)
for off-shell vector bosons. The correspondence between

the HEGET functions and the HELAS codes are shown
explicitly in Tables 3 and 4.

All of the HEGET functions that are used in this re-
port are described in Appendix B.

The massless fermion wave functions with flowing-IN
fermion number, ixxxxk (k=0, 1, 2), and those with flowing-
OUT fermion number, oxxxxk (k = 0, 1, 2), are listed in
Appendix B.1. It is worth noting here that the first 4 com-
ponents of the output complex array fi[6] and fo[6] of
these functions are 4-spinors

|fi> = u(p, nHEL/2) for nSF = +1 (11a)

= v(p, nHEL/2) for nSF = −1 (11b)

<fo| = u(p, nHEL/2) for nSF = +1 (12a)

= v(p, nHEL/2) for nSF = −1 (12b)

just as in the corresponding HELAS subroutines [3]. Al-
though either the first two or the latter two components
vanish for massless fermions, we keep the above generic
4-spinor form because the efficiency gain achievable by
avoiding multiplication of zero’s is significant only for am-
plitudes with a single massless fermion line with many
vector boson emissions and it never exceeds a factor of
two.

The massless vector boson wave functions, vxxxxk (k=
0, 1, 2) are listed in Appendix B.2. Here again the first 4
components of the output complex array vc[6] give the
wave function

(vc) = ǫµ(p, nHEL)∗ for nSV = +1 (13a)

= ǫµ(p, nHEL) for nSV = −1 (13b)

in the light-cone gauge

ǫµnµ = 0 (14)

with the light-like vector

nµ = pµ = (p0,−p1,−p2,−p3) (15)

just like the HELAS subroutines [3] for massless vector
bosons. The use of the light-cone vector (15) allows us
to express the wave functions solely in terms of the four-
momenta of the massless vector bosons.

In Appendix B.3, we list HEGET functions for the FFV
vertex

LF1F2V= ψ
F1
γµ
[

gal[0]
1−γ5
2

+gal[1]
1+γ5
2

]

ψF2
V ∗

µ

(16)

following the HELAS convention [3]. The amplitude func-
tion iovxxx, the off-shell fermion wave functions fvixx0
and fvoxx0, and the off-shell vector wave function jioxx0
are listed in Appendix B.3.1, B.3.2, B.3.3, and B.3.4, re-
spectively. The correspondence between the HEGET func-
tions and the original HELAS subroutines are given in
Table 4.

The complex numbers are defined in each HEGET
functions by including the header file complx.h, given
in Appendix B.4, which has been introduced to handle
complex numbers on CUDA(GPU).

5 Validation of the HEGET functions

We have validated all the HEGET functions by compar-
ing the helicity amplitudes of each process for many phase
space points and for all helicity combinations between
those computed on GPU with the HEGET functions and
those computed on CPU with the FORTRAN version of
HELAS subroutines. For the phase space generation, we
use MadGraph/MadEvent [4] and an independent FOR-
TRAN program which calculates total cross section and
kinematical distributions with the Monte Carlo integra-
tion program BASES [8] as references.

For comparison, we use the same physics parameters
as MadGraph/MadEvent for all programs. As the parton
distribution function, we use CTEQ6L1 [9] and set the
factorization scale to be the pT cut of values, Q = pcut

T
=

20GeV; see eq. (9).

5.1 Total cross sections

For the calculation of the n-photon production cross sec-
tions, the same final state cuts for all the observed photons
are applied; see (eq. (9) in Sec. 2.2). Results for the compu-
tation of the total cross sections are listed in Table 5. We
find the results obtained by the HEGET functions agree
with those from the other programs within the statistics
of generated number of events.

Up to n = 5, MadGraph/MadEvent gives the total
cross sections and distributions, and we find agreements
among all the programs. As for n=6 and 7, the MadGraph
generated HELAS FORTRAN codes can be integrated by
using BASES [8], and the results agree well with those
of HEGET codes. For n = 8, the FORTRAN program
that performs (8!− 1) permutations of eq. (6) was used to
compute the matrix elements, which were integrated by
BASES.

5.2 Kinematical distributions

We also compare several kinematical distributions of the
generated events. As an example, in Fig. 2 distributions
of the maximum transverse momentum and ∆R between
final state photons for the uu → 5-photons process are
shown for the three programs. We find excellent agree-
ment.

 [GeV]
T

p
0 100 200 300 400 500 600 700 8000

0.05

0.1

0.15

0.2

0.25

γ 5→ uu 
 Max.

T
p

Heget

MadGraph
       /MadEvent

Bases

 R∆
0 1 2 3 4 5 6 7 80

0.2

0.4

0.6

0.8

1

1.2

1.4

1.6

1.8

2

γ 5→ uu 

γγ R∆
Heget

MadGraph
       /MadEvent

Bases

Fig. 2. Distributions of maximum transverse momentum and
∆R of final state photons in uu → 5-photons

Similar comparisons have been performed for all the cases
listed in Table 5, and we find agreement within statistical
errors.

6 Performance comparison

6.1 Register allocation and size of the thread block

Event process time is measured by the standard C library
function on the CPU. In order to include the data trans-
fer time between CPU and GPU as an integral part of
the event process time by GPU, we measure the interval
between the time when CPU starts transferring random
numbers to the GPU and the time when the last result
from GPU is received by the CPU. The process time for
one event is obtained by dividing the measured time by the
total number of events. We also prepared a CPU program
of the same structure and measured the event process time
for the equivalent part of the program. The total number
of events used for the measurement depends on the num-
ber of photons in the final state and the specifications of
the GPU. For the GTX280, we generate 1M events for a
single kernel call and repeated 100 calls of the kernel for
processes with nγ≤5. In short, we generated 100M events
for a single execution of the program.

When we compile CUDA programs, the maximum num-
ber of registers allocated to a thread can be specified. How-
ever, the total number of registers available per thread
block is limited. If we allocate more registers to a thread,
then the maximum number of threads in a thread block,
the block size, must become smaller. We find the perfor-
mance of the program depends on these two parameters,
the number of registers in a thread and the number of
threads in a block. Hence, we made a systematic study of
the performance of these parameters.

Table 5. Total cross sections [fb] for uu → nγ at the LHC.

Number of photons HEGET Bases MadGraph/MadEvent

2 1.0822 ± 0.0056 1.08265 ± 0.00031 1.0811 ± 0.0019 ×104

3 6.7776 ± 0.0082 6.7849 ± 0.0051 6.775 ± 0.031 ×100

4 1.2400 ± 0.0030 1.2280 ± 0.0029 1.2372 ± 0.0041 ×10−2

5 2.598 ± 0.011 2.596 ± 0.010 2.572 ± 0.053 ×10−5

6 5.799 ± 0.017 5.792 ± 0.014 — ×10−8

7 1.264 ± 0.008 1.261 ± 0.004 — ×10−10

8 2.77 ± 0.05 2.4± 0.3 — ×10−13

The basic unit of the GPU processor is a multiproces-
sor called the Streaming Multiprocessor (SM). Threads
which belong to one thread block are concurrently exe-
cuted on one SM. One SM consists of eight Scalar Proces-
sors (SP) and four threads can be executed on one SP at
the same time. One multiprocessor can therefore process
a group of 32 threads, called a warp, in parallel.

The number of threads in a thread block of a kernel
execution can be set arbitrarily up to a maximum number
limited by the number of registers per thread, but the
performance is found to be better when it is a multiple of
the warp size, i.e. 32 threads. This is because otherwise
there are idle SPs during the kernel execution.

Fig. 3 shows the dependence of the event process time
for the 5-photon production process, uu→ 5γ, on the size
of thread blocks and the number of allocated registers per
thread on GTX280. GTX280 has 16k registers, which are
split into threads in a thread block. Three cases of num-
bers of registers per thread, 42, 64 and 124, are plotted in
Fig. 3, which correspond to the maximum available num-
bers of threads in a multiple of 32 threads, 384, 256 and
128, respectively.

The dependence on the blocks size shows a clear peri-
odicity of a multiple of 32 threads. The best performance
for the 5γ production process was observed for a combina-
tion of 64 registers per thread and 256 threads in a thread
block.

Number of Threads per Thread Block

Fig. 3. Dependence of event process time on block size (the
number of threads per thread block) and the number of regis-
ters per thread on a GTX280.

6.2 Comparison of the event process time

In Fig. 4, the measured process time for one event of
n-photons production processes is shown for the GPU
(GTX280) and the CPU (PC Linux with Fedora 8). They
are plotted versus the number of photons in the final state.

The upper two lines show the event process time for
the computation on the CPU. The programs with the
MadGraph generated HELAS codes run faster than those
based on the permutation amplitude for larger number of
photons in the final state (nγ ≥ 5). This is essentially be-
cause the MadGraph-generated codes avoid repetition of
computing the same off-shell amplitudes. A larger frac-
tion of off-shell amplitudes is repeated for larger nγ , and
the ratio of the process time for the MadGraph gener-
ated amplitudes and the permutation amplitudes grows
from 1.7 for nγ =5 to 3.8 for nγ =7. For nγ =8, only the
permutation-based program can be compiled on the CPU.

The lower three lines show the event process time on
the GPU (GTX280). In all cases the program on the GPU
computes one event much faster than the CPU. For the
computation on the GPU the HEGET programs obtained
by converting the HELAS codes generated by the Mad-
Graph runs faster than those based on the permutations
of a single amplitude, when nγ = 3 or larger. The ratio
grows to a factor of 2.3 for nγ=5.

When the number of photons becomes larger than five,
the MadGraph amplitude (the HEGET code which is con-
verted from the HELAS code generated by MadGraph)
becomes very long, and the present CUDA compiler can-
not process the converted program. In order to compile the
long HEGET program for large n-photon processes we di-
vide the program into smaller pieces. Each piece computes
a subset of the diagrams. The CPU program calls separate
kernels for each amplitude sequentially and sums up their
computed amplitudes. This method works for n = 7, or
7!=5040 diagrams, whose HEGET code can be compiled
by dividing it into about 500 kernel calls. The 8-photon
production process has so many diagrams (8! = 40320),
that we have not been able to compile the HEGET pro-
gram even after dividing it into small pieces.

For the case of nγ=5 we examine both programs, one
with a single amplitude and the other with divided am-
plitudes. We find that the multiple kernel program with
divided amplitudes runs faster than the single kernel pro-
gram for the whole amplitude by about 30%. This is prob-
ably because the size of the 5γ amplitude program is near
the maximum capability of the present CUDA compiler.

Fig. 4. Event process times for the GPU and CPU.

6.3 Comparison of performance of GPU and CPU

The ratios of event process times between the CPU and
GPU are shown in Fig. 5. The solid curves give the ra-
tios between the permutation amplitudes on the CPU (the
upper CPU points in Fig. 4) and those on the GPU (the
upper GPU points in Fig. 4). The performance ratio is
about 120 for nγ = 2 and 3, and it goes down to about
45 for nγ = 6, 7 and 8. A higher-performance ratio is
found when we compare the event process time of the
MadGraph-generated amplitude on the CPU (the lower
CPU points in Fig. 4) and that on the GPU (the lower
GPU points in Fig. 4). Ratios above 150 are found for
nγ = 3 and 4, going down to about 45 for nγ = 7. For
nγ=6 and 7, the HEGET programs from the MadGraph-
generated HELAS codes are too big for the present CUDA
compiler, and the results are shown for the GPU program
with multiple kernel calls for one event. Nevertheless, the
performance ratio is about 60 for nγ=6 and about 45 for
nγ = 7. Even more surprisingly, the performance is bet-
ter for the divided amplitude with multiple kernel calls
than the single kernel computation for nγ = 5, with the
performance ratio exceeding 100.

To summarize, we find that GPU programs run faster
than CPU programs by a factor exceeding 100 for nγ ≤5,
while a performance gain larger than 40 is achieved for
other nγ≤8.

Fig. 5. Ratio of event process times (CPU/GPU).

6.4 Difference among GPUs

We also examine the performance of two other GPUs,
the 9800GTX and the 8800M GTS. The GPU board with
9800GTX is connected to the same host PC as GTX280.
On the other hand, the 8800M GTS is a built-in graphic
card on the iMac. We ran the same programs for all GPUs
and measures their event process times. Fig. 6 shows the
process time per event for the permutation amplitudes
on the 9800GTX and 8800M GTS relative to a GTX280,
which was shown in Fig. 4 as “GPU permutation”. The
ratio of process times among various GPU’s is roughly
proportional to the inverse of the number of multiproces-
sors: 240 on the GTX280, 128 on the 9800GTX, 64 on
the 8800M GTS; see Table 1. It is worth noting that even
the small built-in graphic card (8800M GTS) on an iMac
can compute nγ production processes faster than CPU by
about factors of 25, 15 and 10, respectively, for nγ =3, 4
and 5.

Fig. 6. Event process times for GPUs.

6.5 Double precision calculation

The GTX280 has one double precision processing unit for
each streaming multiprocessor (SM). We prepared a dou-
ble precision version of the HEGET library and the CUDA
program which computes the nγ cross sections with dou-
ble precision. The structure of these programs is the same
as the single precision version described above. Results of
the single and double precision computations agree. Fig. 7
shows the ratio of event process times between double
and single precision computations. Because the number
of double precision processing units is small compared to
the number of single precision processing units, the pro-
gram with the double precision computations runs 3-4
times slower than that with single precision. In compar-
ison, there is little difference between single and double
precision computation time on CPU.

6.6 Effect of unrolling loops

In the permutation amplitude program, a combination of
integer numbers for successive permutations is generated

sequentially within a while loop of the C-language: see
Appendix A.2 for nγ = 3 as an example. Since programs
with loops or branches lower the efficiency of parallel com-
puting on the GPU, in general, there is a possibility of
improving the performance of the permutation program
by spreading out, or unrolling, a part of the while loop in
the permutation.

We therefore tested the effect of unrolling the while
loop for the permutation generation. Sample code for the
case of the 3-photon production process is shown in
Appendix C as List 17, for the amplitude program with
one permutation unrolled. It should be compared with the
program List 2 in Appendix A.2, which performs 3! = 6
permutation within one while loop. Fig. 8 shows the rela-
tive performance of the programs after unrolling a part of
the while loop as compared to the process time presented
in Fig. 4 as “GPU permutation”.

We find that the effect of unrolling is significant, es-
pecially for large nγ processes. The improvement of the
execution speed becomes a factor of 3 for nγ=7 and 8. Be-
cause the event process time does not change significantly
by unrolling the while loop on the CPU, the CPU/GPU
ratio plotted in Fig. 5 for ‘permutation‘ amplitudes grows
from ∼ 45 to ∼ 150 for nγ =7 and 8 after unrolling two
permutations.

Also plotted as a reference in Fig. 8 are the relative
performances of the computation with the converted Mad-
Graph amplitudes as compared to those of permutation
amplitudes. It is clearly seen from Fig. 8 that by reduc-
ing the while loop the performance of the program with
permutation improves significantly for nγ ≥ 5, approach-
ing that of the MadGraph based programs. The presence
of the while loop is probably the main cause of the poor
performance of the permutation amplitudes as compared
to the MadGraph generated ones, as shown in Fig. 4.

7 Summary

We have presented the results of our attempt to compute
multi-particle production events at hadron colliders on
GPUs [1], Graphic Processing Units. Our achievements
and findings may be summarized as follows.

– HELAS subroutines [3] written in FORTRAN were
converted to HEGET functions written in CUDA [2],
a C-language developed for GPU computing.

– The HELAS amplitude code for uu → nγ’s (nγ ≤ 7)
generated by MadGraph [4] was converted to a CUDA
program which calls HEGET functions.

– CUDA programs that compute (n !−1) permutations of
a single Feynman amplitude were also made for nγ≤8.

– Event process times of the GPU program on a GTX280
are more than 100 times faster than the CPU program
for nγ ≤ 5, while the gain is reduced to about 45 for
nγ=7 and 8.

– The present CUDA compiler cannot process the
HEGET programs converted from the MadGraph gen-
erated HELAS codes for nγ≥6.

– For nγ=5, 6 and 7, we were successful in running the
converted MadGraph programs by dividing the whole
amplitude into smaller pieces and calling them sequen-
tially on the CPU. Performance improves by this divid-
ing procedure for nγ = 5, where the whole amplitude
can be computed by one kernel call.

– CUDA programs based on permutations of one HEGET
amplitude are about a factor of 3 to 4 slower than
those based on divided MadGraph codes for nγ ≥ 5.
The main cause of this slowdown was identified as the
while loop in the program that generates permutation
of nγ-integers.

– The same programs run on the GTX280, 9800GTX
and 8800M GTS, and the performance is roughly pro-
portional to the number of processors: 240, 128 and
64, respectively.

– Double precision amplitudes on a GTX280 were ex-
amined, finding a factor of 2.5 to 4 slower performance
compared to the single precision computation.

Appendix A Sample codes for amplitude
calculations

Appendix A.1 Three photon production amplitude
with the converted MadGraph code

List 1. uux3a(madgraph).cu

cmplx w01[6], w02[6], w03[6], w04[6], w05[6];

ixxxx1(p1, nh1, +1, w01);
oxxxx2(p2, nh2, -1, w02);
vxxxx0(p3, nh3, +1, w03);
vxxxx0(p4, nh4, +1, w04);
vxxxx0(p5, nh5, +1, w05);

cmplx w06[6], w07[6], w08[6];

cmplx ampsum = mkcmplx(0.0f, 0.0f);
cmplx amp;

fvoxx0(w02,w03,gau,w06);
fvoxx0(w06,w04,gau,w07);
iovxxx(w01,w07,w05,gau,amp);
ampsum += amp;
fvixx0(w01,w04,gau,w07);
fvoxx0(w02,w05,gau,w08);
iovxxx(w07,w08,w03,gau,amp);
ampsum += amp;
fvoxx0(w02,w03,gau,w06);
fvixx0(w01,w04,gau,w07);
iovxxx(w07,w06,w05,gau,amp);
ampsum += amp;
fvoxx0(w02,w04,gau,w06);
fvixx0(w01,w05,gau,w07);
iovxxx(w07,w06,w03,gau,amp);
ampsum += amp;
fvixx0(w01,w03,gau,w07);
fvixx0(w07,w04,gau,w08);
iovxxx(w08,w02,w05,gau,amp);
ampsum += amp;
fvixx0(w01,w03,gau,w07);
fvoxx0(w02,w04,gau,w06);
iovxxx(w07,w06,w05,gau,amp);
ampsum += amp;

Appendix A.2 Three photon production amplitude
with permutations

List 2. uux3a(permutation).cu

cmplx w[5][6];

ixxxx1(p1, nh1, +1, w[0]);
oxxxx2(p2, nh2, -1, w[1]);
vxxxx0(p3, nh3, +1, w[2]);
vxxxx0(p4, nh4, +1, w[3]);
vxxxx0(p5, nh5, +1, w[4]);

cmplx w01[6],w02[6];

cmplx ampsum = mkcmplx(0.0f, 0.0f);

int Flag_Perm=1;
int a[3];
a[0] = 2;
a[1] = 3;
a[2] = 4;

do {
cmplx amp;
fvoxx0(w[1], w[a[0]], gau, w01);
fvoxx0(w01 , w[a[1]], gau, w02);
iovxx0(w[0], w02, w[a[2]], gau, amp);
ampsum += amp;
iPNext(a, 3, Flag_Perm);

} while (Flag_Perm>0);

Appendix B HEGET functions

All of the HEGET functions that are used in this report
are explained in this Appendix.

Appendix B.1 Fermion wave functions

We have two types of functions to compute external fermi-
ons. One is for “flowing-In” fermions and the other is for
“flowing-Out” fermions.

For the flowing-In spinor wavefunction of a fermion,
we prepare three functions, ixxxx1 (List 3), ixxxx2 (List
4) and ixxxx0 (List 5), derived from the HELAS sub-
routine, IXXXXX; see Table 3 for correspondence between
the HEGET functions and the HELAS subroutines [3].
These three functions compute massless fermion wave-
functions with the flowing-In fermion number. ixxxx1 is
for a fermion with momentum along the +z direction,
ixxxx2 is for a fermion with momentum along the −z di-
rection, and ixxxx0 is for a fermion with a generic three-
momentum, p. All functions have the same argument list:

ixxxxk(float* p, int nHEL, int nSF, cmplx* fi)

(17)

for k = 1, 2 and 0. The inputs and outputs are:

Inputs:

float p[4] 4-momentum
int nHEL twice fermion helicity (-1 or 1)
int nSF +1 for particle, -1 for anti-particle

Outputs:

cmplx fi[6] fermion wavefunction |fi>
u(p, nHEL/2) for nSF = +1
v(p, nHEL/2) for nSF = −1

(18)

For the flowing-Out spinor wavefunction of a fermion,
we also prepare three functions, oxxxx1 (List 6), oxxxx2
(List 7) and oxxxx0 (List 8), derived from the HELAS
subroutine, OXXXXX; see Table 3.

These three functions compute a massless fermion wave-
function with a flowing-Out fermion number. oxxxx1 is for
a fermion with momentum along the +z direction, oxxxx2
is for a fermion with momentum along −z direction, and
oxxxx0 is for a final state fermion with a generic three-
momentum, p. All functions have the same argument list:

oxxxxk(float* p, int nHEL, int nSF, cmplx* fo)

(19)

for k = 1, 2 and 0. The inputs and outputs are:

Inputs:

float p[4] 4-momentum
int nHEL twice fermion helicity (-1 or 1)
int nSF +1 for particle, -1 for anti-particle

Outputs:

cmplx fo[6] fermion wavefunction <fo|
u(p, nHEL/2) for nSF = +1
v(p, nHEL/2) for nSF = −1

(20)

List 3. ixxxx1.cu

#include "cmplx.h"

__device__
void ixxxx1(float* p, int nHEL, int nSF, cmplx* fi)
{

float SQP0P3 = sqrtf(p[0]+p[3])*(float)(nSF);
int NH = nHEL*nSF;

fi[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fi[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx(NH*p[1]*(1.0f/SQP0P3),
p[2]*(1.0f/SQP0P3));

cmplx CZERO = mkcmplx(0.0f, 0.0f);
cmplx CSQP0P3 = mkcmplx(SQP0P3, 0.0f);

fi[0]=(NH== 1)*CZERO + (NH==-1)*CHI;
fi[1]=(NH== 1)*CZERO + (NH==-1)*CSQP0P3;
fi[2]=(NH== 1)*CSQP0P3 + (NH==-1)*CZERO;
fi[3]=(NH== 1)*CHI + (NH==-1)*CZERO;

return;
}

List 4. ixxxx2.cu

#include "cmplx.h"

__device__
void ixxxx2(float* p, int nHEL, int nSF, cmplx* fi)
{

int NH=nHEL*nSF;

fi[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fi[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx( nHEL*sqrtf(2.0f*p[0]),
0.0f);

cmplx CZERO = mkcmplx(0.0f,0.0f);

fi[0]=(NH== 1)*CZERO + (NH==-1)*CHI;
fi[1]=CZERO;
fi[2]=CZERO;
fi[3]=(NH== 1)*CHI + (NH==-1)*CZERO;

return;
}

List 5. ixxxx0.cu

#include "cmplx.h"

__device__
void ixxxx0(float* p, int nHEL, int nSF, cmplx* fi)
{

float SQP0P3 = sqrtf(p[0]+p[3])*(float)(nSF);
int NH = nHEL*nSF;

fi[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fi[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx( NH*p[1]*(1.0f/SQP0P3),
(1.0f/SQP0P3)*p[2]);

cmplx CZERO = mkcmplx(0.0f,0.0f);
cmplx CSQP0P3 = mkcmplx(SQP0P3, 0.0f);

fi[0]=(NH== 1)*CZERO + (NH==-1)*CHI;
fi[1]=(NH== 1)*CZERO + (NH==-1)*CSQP0P3;
fi[2]=(NH== 1)*CSQP0P3 + (NH==-1)*CZERO;
fi[3]=(NH== 1)*CHI + (NH==-1)*CZERO;

return;
}

List 6. oxxxx1.cu

#include "cmplx.h"

__device__
void oxxxx1(float* p, int nHEL, int nSF, cmplx* fo)
{

float SQP0P3=sqrtf(p[0]+p[3])*(float)(nSF);
int NH=nHEL*nSF;

fo[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fo[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx( NH*p[1]*(1.0f/SQP0P3),
-p[2]*(1.0f/SQP0P3));

cmplx CZERO = mkcmplx(0.0f, 0.0f);
cmplx CSQP0P3 = mkcmplx(SQP0P3, 0.0f);

fo[0]=(NH== 1)*CSQP0P3 + (NH==-1)*CZERO;
fo[1]=(NH== 1)*CHI + (NH==-1)*CZERO;
fo[2]=(NH== 1)*CZERO + (NH==-1)*CHI;
fo[3]=(NH== 1)*CZERO + (NH==-1)*CSQP0P3;

return;
}

List 7. oxxxx2.cu

#include "cmplx.h"

__device__
void oxxxx2(float* p, int nHEL, int nSF, cmplx* fo)
{

int NH=nHEL*nSF;

fo[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fo[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx(-nHEL*sqrtf(2.0f*p[0]),
0.0f);

cmplx CZERO = mkcmplx(0.0f,0.0f);

fo[0]=CZERO;
fo[1]=(NH== 1)*CHI + (NH==-1)*CZERO;
fo[2]=(NH== 1)*CZERO + (NH==-1)*CHI;
fo[3]=CZERO;

return;
}

List 8. oxxxx0.cu

#include "cmplx.h"

__device__
void oxxxx0(float* p, int nHEL, int nSF, cmplx* fo)
{

float SQP0P3 = sqrtf(p[0]+p[3])*(float)(nSF);
int NH = nHEL*nSF;

fo[4] = mkcmplx(p[0]*(float)(nSF),
p[3]*(float)(nSF));

fo[5] = mkcmplx(p[1]*(float)(nSF),
p[2]*(float)(nSF));

cmplx CHI = mkcmplx( NH*p[1]*(1.0f/SQP0P3),
-p[2]*(1.0f/SQP0P3));

cmplx CZERO = mkcmplx(0.0f, 0.0f);
cmplx CSQP0P3 = mkcmplx(SQP0P3, 0.0f);

fo[0]=(NH== 1)*CSQP0P3 + (NH==-1)*CZERO;
fo[1]=(NH== 1)*CHI + (NH==-1)*CZERO;
fo[2]=(NH== 1)*CZERO + (NH==-1)*CHI;
fo[3]=(NH== 1)*CZERO + (NH==-1)*CSQP0P3;

return;
}

Appendix B.2 Vector boson wave functions

For the wavefunction of a massless vector boson, we pre-
pare three functions, vxxxx1 (List 9), vxxxx2 (List 10)
and vxxxx0 (List 11), derived from the HELAS subrou-
tine, VXXXXX; see Table 3.

These three functions compute a massless vector bo-
son wavefunction from its four-momentum and helicity.
vxxxx1 is for a vector boson with momentum along the
+z direction, vxxxx2 is for a vector boson with momentum
along −z direction, and vxxxx0 is for a vector boson with
a generic three-momentum, p. Their argument list is:

vxxxxk(float* p, int nHEL, int nSV, cmplx* vc)

(21)

for k = 1, 2 and 0. The inputs and outputs are:

Inputs:

float p[4] 4-momentum
int nHEL helicity of massless vector (-1 or 1)
int nSV +1 for final, -1 for initial vector

Outputs:

cmplx vc[6] vector boson wavefunction
ǫµ(p, nHel)∗ for nSV = +1
ǫµ(p, nHel) for nSV = −1.

(22)

List 9. vxxxx1.cu

#include "cmplx.h"

__device__
void vxxxx1(float* p, int nHEL, int nSV, cmplx* vc)
{

vc[4] = mkcmplx(-p[0], -p[3]);
vc[5] = mkcmplx(-p[1], -p[2]);

vc[0] = mkcmplx(0.0f, 0.0f);
vc[3] = mkcmplx(0.0f, 0.0f);

vc[1] = mkcmplx(-(float)(nHEL)*rSQRT2, 0.0f);
vc[2] = mkcmplx( 0.0f, -rSQRT2);

return;
}

List 10. vxxxx2.cu

#include "cmplx.h"

__device__
void vxxxx2(float* p, int nHEL, int nSV, cmplx* vc)
{

vc[4] = mkcmplx(-p[0], -p[3]);
vc[5] = mkcmplx(-p[1], -p[2]);

vc[0] = mkcmplx(0.0f, 0.0f);
vc[3] = mkcmplx(0.0f, 0.0f);

vc[1] = mkcmplx(-(float)(nHEL)*rSQRT2, 0.0f);
vc[2] = mkcmplx( 0.0f, rSQRT2);

return;
}

List 11. vxxxx0.cu

#include "cmplx.h"

__device__
void vxxxx0(float* p, int nHEL, int nSV, cmplx* vc)
{

vc[4] = mkcmplx(p[0], p[3])*nSV;
vc[5] = mkcmplx(p[1], p[2])*nSV;

float rpt = rsqrtf(p[1]*p[1] + p[2]*p[2]);

vc[0] = mkcmplx(0.0f, 0.0f);
vc[3] = mkcmplx(

(float)(nHEL)*(1.0f/(rpt*p[0]))*rSQRT2,
0.0f);

float pzpt = (p[3]*(1.0f/p[0])*rpt)*rSQRT2
*(float)(nHEL);

vc[1] = mkcmplx(-p[1]*pzpt,
-nSV*p[2] * rpt * rSQRT2);

vc[2] = mkcmplx(-p[2]*pzpt,
+nSV*p[1] * rpt * rSQRT2);

return;
}

Appendix B.3 Fermion-fermion-vector boson vertex

For the fermion-fermion-vector boson vertex,

LF1F2V= ψF1
γµ
[

gal[0]
1−γ5
2

+gal[1]
1+γ5
2

]

ψF2
V ∗

µ

(23)

there are four functions, iovxxx, fvixx0, fvoxx0 and
jioxx0, in HEGET. They correspond to HELAS subrou-
tines, IOVXXX, FVIXXX, FVOXXX and JIOXXX, respectively,
for massless particles; see Table 4 for the correspondence
between the HEGET functions and the HELAS subrou-
tines [3].

Appendix B.3.1 iovxxx

The function iovxxx (List 12) computes the amplitude of
the FFV vertex from a flowing-In fermion, a flowing-Out
fermion and a vector boson wave functions, whether they
are on-shell or off-shell. It has the arguments:

iovxxx(cmplx* fi, cmplx* fo, cmplx* vc,
float* gal, cmplx vertex)

(24)

where the inputs and the outputs are:

Inputs:

cmplx fi[6] flowing-In fermion wavefunction
cmplx fo[6] flowing-Out fermion wavefunction
cmplx vc[6] vector wavefunction
float gal[2] coupling constants of FFV vertex

Outputs:

cmplx vertex amplitude of the FFV vertex
<fo|V|fi>

(25)

The two chiral couplings of eq. (25) are:

gal[0] = gal[1] = −eQq (26)

for the qqγ vertex in QED, where Qq is the electric charge

of the quark in units of the proton charge, e =
√
4πα.

Appendix B.3.2 fvixx0

The function fvixx0 (List. 13) computes the off-shell mass-
less fermion wavefunction from a flowing-In external fermi-
on and a vector boson. It has the arguments:

fvixx0(cmplx* fi, cmplx* vc, float* gal,
cmplx* fvi)

(27)

where the inputs and the outputs are:

Inputs:

cmplx fi[6] flowing-In fermion wavefunction
cmplx vc[6] vector wavefunction
float gal[2] coupling constants of the FFV vertex

Outputs:

cmplx fvi[6] off-shell fermion wavefunction
|f’,vc,fi>

(28)

Appendix B.3.3 fvoxx0

The function fvoxx0 (List. 14) computes the off-shell mass-
less fermion wavefunction from a flowing-Out external fermi-
on and a vector boson. It has the arguments:

fvoxx0(cmplx* fo, cmplx* vc, float* gal,
cmplx* fvo)

(29)

where the inputs and the outputs are:

Inputs:

cmplx fo[6] flowing-Out fermion wavefunction
cmplx vc[6] vector wavefunction
float gal[2] coupling constants of the FFV vertex

Outputs:

cmplx fvo[6] off-shell fermion wavefunction
<fo,vc,f’|

(30)

Appendix B.3.4 jioxx0

This function, jioxx0 (List 15) computes the off-shell vec-
tor wavefunction from an external fermion pair. The off-
shell vector boson wavefunction is given in the Feynman
gauge for massless vectors (photons and gluons in the SM).
It has the arguments:

jioxx0(cmplx* fi, cmplx* fo, float* gal,
cmplx* jio)

(31)

where the inputs and the outputs are:

Inputs:

cmplx fi[6] flowing-In fermion wavefunction
cmplx fo[6] flowing-Out fermion wavefunction
float gal[2] coupling constants of the FFV vertex

Outputs:

cmplx jio[6] vector current jµ (<fo|V|fi>)

(32)

List 12. iovxxx.cu

#include "cmplx.h"

__device__
void iovxx0(cmplx* fi, cmplx* fo, cmplx* vc,

float* gal, cmplx& vertex)
{

vertex =
gal[0]*((fo[2]*fi[0]+fo[3]*fi[1])*vc[0]

+(fo[2]*fi[1]+fo[3]*fi[0])*vc[1]
-((fo[2]*fi[1]-fo[3]*fi[0])*vc[2])
*mkcmplx(0.0f, 1.0f)
+(fo[2]*fi[0]-fo[3]*fi[1])*vc[3])

+gal[1]*((fo[0]*fi[2]+fo[1]*fi[3])*vc[0]
-(fo[0]*fi[3]+fo[1]*fi[2])*vc[1]
+((fo[0]*fi[3]-fo[1]*fi[2])*vc[2])
*mkcmplx(0.0f, 1.0f)
-(fo[0]*fi[2]-fo[1]*fi[3])*vc[3]);

return;
}

List 13. fvixx0.cu

#include "cmplx.h"

__device__
void fvixx0(cmplx* fi, cmplx* vc, float* gal,

cmplx* fvi)
{

fvi[4] = fi[4]-vc[4];
fvi[5] = fi[5]-vc[5];

float pf[4];
pf[0] = fvi[4].re;
pf[1] = fvi[5].re;
pf[2] = fvi[5].im;
pf[3] = fvi[4].im;

float pf2 = pf[0]*pf[0] -
(pf[1]*pf[1] + pf[2]*pf[2] + pf[3]*pf[3]);

cmplx cI = mkcmplx( 0.0f, 1.0f);
float d = -1.0f/pf2;

cmplx sl1 = (vc[0] + vc[3])*fi[0]
+ (vc[1] - cI*vc[2])*fi[1];

cmplx sl2 = (vc[0] - vc[3])*fi[1]
+ (vc[1] + cI*vc[2])*fi[0];

cmplx sr1 = (vc[0] - vc[3])*fi[2]
- (vc[1] - cI*vc[2])*fi[3];

cmplx sr2 = (vc[0] + vc[3])*fi[3]
- (vc[1] + cI*vc[2])*fi[2];

fvi[0] = ( gal[0]*((pf[0]-pf[3])*sl1
- conj(fvi[5])*sl2))*d;

fvi[1] = ( gal[0]*((pf[0]+pf[3])*sl2
- fvi[5] *sl1))*d;

fvi[2] = ( gal[1]*((pf[0]+pf[3])*sr1
+ conj(fvi[5])*sr2))*d;

fvi[3] = ( gal[1]*((pf[0]-pf[3])*sr2
+ fvi[5] *sr1))*d;

return;
}

List 14. fvoxx0.cu

#include "cmplx.h"

__device__
void fvoxx0(cmplx* fo, cmplx* vc, float* gal,

cmplx* fvo)
{

fvo[4] = fo[4]+vc[4];
fvo[5] = fo[5]+vc[5];

float pf[4];
pf[0] = fvo[4].re;
pf[1] = fvo[5].re;
pf[2] = fvo[5].im;
pf[3] = fvo[4].im;

float pf2 = pf[0]*pf[0]
- (pf[1]*pf[1] + pf[2]*pf[2] + pf[3]*pf[3]);

cmplx cI = mkcmplx( 0.0f, 1.0f);

float d = -1.0f/pf2;

cmplx sl1 = (vc[0] + vc[3])*fo[2]
+ (vc[1] + cI*vc[2])*fo[3];

cmplx sl2 = (vc[0] - vc[3])*fo[3]
+ (vc[1] - cI*vc[2])*fo[2];

cmplx sr1 = (vc[0] - vc[3])*fo[0]
- (vc[1] + cI*vc[2])*fo[1];

cmplx sr2 = (vc[0] + vc[3])*fo[1]
- (vc[1] - cI*vc[2])*fo[0];

fvo[0] = ( gal[1]*((pf[0]+pf[3])*sr1
+ fvo[5] *sr2 ))*d;

fvo[1] = ( gal[1]*((pf[0]-pf[3])*sr2
+ conj(fvo[5])*sr1 ))*d;

fvo[2] = ( gal[0]*((pf[0]-pf[3])*sl1
- fvo[5] *sl2 ))*d;

fvo[3] = ( gal[0]*((pf[0]+pf[3])*sl2
- conj(fvo[5])*sl1 ))*d;

return;
}

List 15. jioxx0.cu

#include "cmplx.h"

__device__
void jioxx0(cmplx* fi, cmplx* fo, float* gal,

cmplx* jio)
{

jio[4]=fo[4]-fi[4];
jio[5]=fo[5]-fi[5];

float DD=1.0f/(
(jio[4].re)*(jio[4].re) - (jio[5].re)*(jio[5].re)

- (jio[5].im)*(jio[5].im) - (jio[4].im)*(jio[4].im));

jio[0]=(gal[0]*(fo[2]*fi[0]+fo[3]*fi[1])
+gal[1]*(fo[0]*fi[2]+fo[1]*fi[3]))*DD;

jio[1]=(gal[1]*(fo[0]*fi[3]+fo[1]*fi[2])
-gal[0]*(fo[2]*fi[1]+fo[3]*fi[0]))*DD;

jio[2]=(gal[0]*(fo[2]*fi[1]-fo[3]*fi[0])
-gal[1]*(fo[0]*fi[3]-fo[1]*fi[2]))

*mkcmplx(0.0f,DD);

jio[3]=(gal[1]*(fo[0]*fi[2]-fo[1]*fi[3])
-gal[0]*(fo[2]*fi[0]-fo[3]*fi[1]))*DD;

return;
}

Appendix B.4 Complex numbers

List 16. cmplx.h

#ifndef __cmplx_h__
#define __cmplx_h__

typedef struct __align__(8){
float re;
float im;

} cmplx;

inline __host__ __device__
cmplx mkcmplx(float re, float im){

cmplx z;
z.re=re;
z.im=im;
return z;

}

inline __host__ __device__
cmplx mkcmplx(cmplx z){

return mkcmplx(z.re,z.im);
}

inline __host__ __device__
cmplx mkcmplx(float a){

return mkcmplx(a,0.0);
}

inline __host__ __device__

float real(cmplx a){
return a.re;

}

inline __host__ __device__
float imag(cmplx a){

return a.im;
}

inline __host__ __device__
cmplx conj(cmplx a){

return mkcmplx(a.re,-a.im);
}

inline __host__ __device__
cmplx operator+(cmplx a, cmplx b){

return mkcmplx(a.re + b.re, a.im + b.im);
}

inline __host__ __device__
void operator+=(cmplx &a, cmplx b){

a.re += b.re; a.im += b.im;
}

inline __host__ __device__
cmplx operator+(cmplx a){

return mkcmplx(+a.re, +a.im);
}

inline __host__ __device__
cmplx operator-(cmplx a, cmplx b){

return mkcmplx(a.re - b.re, a.im - b.im);
}

inline __host__ __device__
void operator-=(cmplx &a, cmplx b){

a.re -= b.re; a.im -= b.im;
}

inline __host__ __device__
cmplx operator-(cmplx a){

return mkcmplx(-a.re, -a.im);
}

inline __host__ __device__
cmplx operator*(cmplx a, cmplx b){

return mkcmplx((a.re * b.re) - (a.im * b.im),
(a.re * b.im) + (a.im * b.re));

}

inline __host__ __device__
cmplx operator*(cmplx a, float s){

return mkcmplx(a.re * s, a.im * s);
}

inline __host__ __device__
cmplx operator*(float s, cmplx a){

return mkcmplx(a.re * s, a.im * s);
}

inline __host__ __device__
void operator*=(cmplx &a, float s){

a.re *= s; a.im *= s;
}

inline __host__ __device__
cmplx operator/(cmplx a, cmplx b){

float tmpD=(1.0f/(b.re*b.re+b.im*b.im));
return mkcmplx(

( (a.re * b.re) + (a.im * b.im))*tmpD,
(-(a.re * b.im) + (a.im * b.re))*tmpD
);

}

inline __host__ __device__
cmplx operator/(cmplx a, float s){

float inv = 1.0f / s;
return a * inv;

}

inline __host__ __device__
cmplx operator/(float s, cmplx a){

float inv = s*(1.0f/(a.re*a.re+a.im*a.im));
return mkcmplx(inv*a.re,-inv*a.im);

}

inline __host__ __device__
void operator/=(cmplx &a, float s){

float inv = 1.0f / s;
a *= inv;

}

inline __host__ __device__

float fabsc(cmplx a){
return sqrtf((a.re*a.re)+(a.im*a.im));

}

inline __host__ __device__
float fabs2c(cmplx a){

return (a.re*a.re)+(a.im*a.im);
}
#endif

Appendix C A sample code for unrolling the
permutation loop for three photon
production process

List 17. uux3a(unrolled).cu

cmplx w[5][6];

ixxxx1(p1, nh1, +1, w[0]);
oxxxx2(p2, nh2, -1, w[1]);
vxxxx0(p3, nh3, +1, w[2]);
vxxxx0(p4, nh4, +1, w[3]);
vxxxx0(p5, nh5, +1, w[4]);

cmplx w01[6],w02[6];

cmplx ampsum = mkcmplx(0.0f, 0.0f);

int Flag_Perm=1;
int a[3];
a[0] = 2;
a[1] = 3;
a[2] = 4;

do {
cmplx amp;
fvoxx0(w[1], w[a[0]], gau, w09);
fvoxx0(w09 , w[a[1]], gau, w10);
iovxx0(w[0], w10, w[a[2]], gau, amp);
ampsum += amp;
fvoxx0(w[1], w[a[0]], gau, w09);
fvoxx0(w09 , w[a[2]], gau, w10);
iovxx0(w[0], w10, w[a[1]], gau, amp);
ampsum += amp;
fvoxx0(w[1], w[a[2]], gau, w09);
fvoxx0(w09 , w[a[0]], gau, w10);
iovxx0(w[0], w10, w[a[1]], gau, amp);
ampsum += amp;
iPNext(a, 2, Flag_Perm);

} while (Flag_Perm>0);

